



Criando um calendário mensal

por **Luiz Osório Paim Pereira**

Nesta dica, vamos criar um calendário mensal permanente, usando as funções **SE**, **DIA.-DA.SEMANA**, **PROCURAR**, **DIA**, **OU** e **DIASNOMÊS**, do **Calc**.

Eis uma figura do calendário para o mês de Janeiro de 2007:

	B	C	D	E	F	G	H
1	Jan 2007		Calendário				
2	Dom		7	14	21	28	
3	Seg	1	8	15	22	29	
4	Ter	2	9	16	23	30	
5	Qua	3	10	17	24	31	
6	Qui	4	11	18	25		
7	Sex	5	12	19	26		
8	Sab	6	13	20	27		

Na célula **B1** entramos com a data do primeiro dia do mês. A seguir, aplicamos o formato **MMM AAAA** comandando:

Formatar > Células > Número > Data > Dez 1999

Nas células do intervalo **B2:B8** entramos com as abreviações dos dias da semana. No intervalo **C2:H8** calculamos os dias do mês conforme abaixo.

Nas células **C2** a **C8** temos o cálculo da primeira semana do mês/ano em questão com a fórmula:

```
=SE(C1<>"";C1+1;SE(DIA.DA.SEMANA($B$1)=((PROCURAR(B2;"DomSegTerQuaQuiSexSab";1)-1)/3)+1;DIA($B$1);""))
```

Começamos testando a célula acima para saber se não está vazia. Caso a condição seja verdadeira (não vazia), incrementamos em 1 para obter o próximo dia. A parte da fórmula envolvida é:

```
SE(C1<>"";C1+1; ...
```

Caso a condição seja falsa (célula acima vazia), testamos para saber se o primeiro dia do mês do calendário é igual ao dia da semana da coluna anterior. A parte da fórmula envolvida no teste é:

```
... SE(DIA.DA.SEMANA($B$1)=((PROCURAR(B2;"DomSegTerQuaQuiSexSab";1)-1)/3)+1;DIA($B$1);""))
```

Para tal, usamos um pequeno algoritmo, apoiado na função **PROCURAR**, para comparar com o resultado da função **DIA.DA.SEMANA(\$B\$1)**.

PROCURAR(o valor que está na col B;na string "DomSegTerQuaQuiSexSab";a partir da posição 1 da dita string).

Eis um resumo das operações para o primeiro dia do mês em **B1**:

Nome do dia	Procurar	Operações	Dia.Da.Semana
=====	=====	=====	=====
Dom	1	$(1-1)/3 + 1 = 1$	2
Seg	4	$(4-1)/3 + 1 = 2$	2 <= dia primeiro: Segunda-feira
Ter	7	$(7-1)/3 + 1 = 3$	2
Qua	10	$(10-1)/3 + 1 = 4$	2
Qui	13	$(13-1)/3 + 1 = 5$	2
Sex	16	$(16-1)/3 + 1 = 6$	2
Sáb	19	$(19-1)/3 + 1 = 7$	2


Após definir a primeira semana, o próximo passo é calcular os valores das outras semanas.

Em **D2**, **E2**, **F2**, **G2** e **H2**, testamos a última célula da coluna anterior para saber se é vazia ou se, incrementando seu conteúdo em 1, resulta num valor maior que o número de dias do mês. Neste caso, encerramos. Caso contrário, o valor da célula inicial da semana será igual ao da última célula da semana anterior acrescido em 1. Para isto, entre com a fórmula abaixo no intervalo **D2:H2**, ajustando a última célula:

```
=SE(OU(C8="";C8+1>DIASNOMÊS($B$1));"";C8+1)
```

De **D3** a **H8**, também usamos as funções **SE**, **OU** e **DIASNOMÊS**, só que agora testando a célula acima e não a última célula da coluna anterior. A fórmula é a mesma, com uma pequena alteração:

```
=SE(OU(D2="";D2+1>DIASNOMÊS($B$1));"";D2+1)
```

Neste exemplo, mostramos o poder das funções do **Calc**, usadas individualmente ou combinadas, para solucionar problemas comuns num escritório. A planilha com o calendário pode ser baixada pelos leitores na página do *Concurso de Dicas do BrOffice.org*[1]. 

Referência:

[1] <http://broffice.org/concurso>